
 بسم االله السميع العليم

   في برامجناRSSاستغلال الـ 

 

 RSS، و لذا فلا تساورني اي شكوك في أن آلمة مثل ) مطور برامج او مواقع ويب (بما انك تقرأ آلماتي هذه فأنت بالطبع مبرمج 

( ر الان، انظر الى هذه ايضا  تتذآأظنك ) ( قد مرت عليك اثناء تصفك للإنترنت و لشحن ذاآرتك قليلا انظر الى هذه الايقونة 

  ... (  

Really Simple Syndication 

اذا تصفحت النت و المنتديات فستجد الكثيرين قد تحدثوا عنها بإسهاب و لا اخفيك انني لا اريد ان اآون واحدا منهم اي انني لا 
  انوي الإسهاب لانهم قد غطوا الموضوع بجوانبه و الحمد الله 

  ؟، أمثله؟، متى استغلها آمستخدم؟ متى استغلها آمزود؟عنهاائك فكرة سريعة ما اريده هنا هو إعط

  مقدمة

 فلو اردتني ان البياناتلوصف  ما هي إلا لغة XML و الـ XML عبارة بكل اختصار عن نصوص RSSالـ ... دعنا نتوآل على االله و نبدأ 
  : فسأقولXMLاصف نفسي بالـ 

<Person> 

 <Name>Ragheed Al-Tayeb</Name> 

 <Age>27 Years</Age> 

 <Job>IT Engineer</Job> 

</Person> 

  

  

  

  

  

 

 ,Name ثم وضعت فيه عناصر فرعية بإسم Person الشخصية في عنصر بإسم ي وضعت معلوماتلو تمعنت فيما سبق ستجد انني
Age, Job  تقراها على  وهي تصف اسمي و عمري و عملي و اذا ارسلت لك ملف يحتوي على هذه البيانات فأنت بسهولة سوف

 و التي تعتمد XMLاصفه لك لأنها ببساطة لغة الـ الرغم من عدم وجود اتفاق مسبق بيننا و بنظرة خاطفة عليها ستفهم ما اريد ان 
) رغيد ( على البساطة في الوصف و الذي يعتمد بدوره على مدى بساطة الكائن المراد وصفه و الذي هو في مثالنا السابق 

  .و الحمد الله فوصفي جاء بسيط) غير معقد (ولأنني آائن بسيط

  : لاحظ

تزيد او تنقص اعتمادا على الاشياء التي يريد المصدر وصفها فمثلا لو سألت زيدا ان يصف اي ان !  هذه العناصر يمكن ان تختلف ان
ولربما استبدل )  Person, Name, Age & Job(ان تجده قد استخدم نفس العناصر التي استخدمتها انا ) ليس بالضروري(نفسه 

<Man> بدلا عن <Person> و لربما اضاف عنصر مثل Mobile او ربما e-Mail و استغنى عن وصف عمله Job بالتالي يكون الملف و
  :الذي آونه قريب من التالي 

<Man> 

 <Name>Zaid Sameh</Name> 

 <Age>20 Years</Age> 

 <Mobile>123456789</Mobile> 

 <e-mail>zaid@somewhere.com</e-mail> 

</Man> 

 

  

  

  

  

  

  

  

  



 

تحتوي هذه العناصر على خصائص معينة فمثلا اذا آان هناك خصائص خاصة بعنصر معين فقد اضافة بسيطة و هي انه يمكن ان 
 اي يصف Jobالـ ذا اذا اردت ان اذآر شيئا يخص هذا وهو عنصر يصفني فما) العمل( Jobقمت في مثالي الاول بذآر عنصر بإسم 

 وهي Attributesفهنا تظهر الحاجة الى الخصائص ) ...  خاص او حكومي :مثل نوع العمل(هذا العمل و لا يصف رغيد بحد ذاته 
  :خذ المثال التالي ... ذاته عنصر بالعبارة صفات خاصة 

  

  

  

  

  

  

<Person> 

 <Name>Ragheed Al-Tayeb</Name> 

 <Age>27 Years</Age> 

 <Job JobType=”Private”>IT Engineer</Job> 

</Person> 

  . التي تدل على آون العمل هو عمل خاصPrivate و اسندت القيمة JobTypeالخاصية وفيه ترى بوضوح انني اضفت 

 لغة وحدت تبادل البيانات و XMLلغة الـ (   XML عن الـ و الان اذا آان رأسك قد آلمك من قبل من العبارات التي يكتبها امثالي
ها الى قاعدة اخرى مختلفة عنها عن طريق الـ يعني قاعدة بيانات من نوع معين تنقل بيانات) سهلت تناقله بين المصادر المختلفة 

XML و ... و ...  وو الكثير الكثير  لغة برمجة تتفاهم مع اخرى عن طريقها ايضا!!! ...  

   ..!رغيد وصف نفسه بطريقة و زيد وصف نفسه بأخرى؟طالما أن  جاء هذا التوحد الذي يتكلمون عنه نفمن أي

نهم جميعا على عناصر وخصائص موحدة و بالتالي فعلى آل مصدر ان يوصف نفسه هذا في الحقيقة يأتي من وجود اتفاق بي
 آلها تخزن و mySQL او Access او Oracle او SQL Serverبإستخدام نفس العناصر المتفق عليها فمثلا جدول في قاعدة بيانات 

 فما الجدول إلا عبارة عن XMLان تصف نفسها بالـ تتعامل مع الجداول و العلاقات بينها بطرق مختلفة و لكن اذا ارادت هذه اللغات 
حقول و سجلات و بالتالي اذا تم الاتفاق على مجموعة من العناصر و خصائصها بينهم فمن السهل ان تصف قاعدة بيانات آاملة 

قاعدة بيانات موصوفة  و بالمثال يتضح المقال آما يقول اخي العزيز الغملاسي دائما، لهذا سأذآر مثلا بسيطا لXMLبإستخدام الـ 
  :آما يلي  ... XMLبالـ 

 
<Database Name=”SchoolData”> 

 <Table Name=”Students” FieldsCount=3> 

  <Row> 

   <Name Type=”Text” Length=20>Ragheed</Name> 

   <Age Type=”Number” Length=2>27</Age> 

   <Male Type=”Boolean”>True</Male> 

  </Row> 

  <Row> 

   <Name Type=”Text” Length=20>Zaid</Name> 

   <Age Type=”Number” Length=2>20</Age> 

   <Male Type=”Boolean”>True</Male> 

  </Row> 

 </Table> 

</Database> 

 

 

 

 

 

  

  

  

  

  

  

  

  



 و انها تحتوي على جدول واحد SchoolDataيالها من سهولة فببساطة تستطيع ان تعرف ان قاعدة البيانات السابقة هي بإسم 
ر الى هذه العناصر تجد ان  و بالنظMale و Age و  Name ثلاثة حقول هي FieldsCount=3 وهذا الجدول مكون من Studentsبإسم 
 True اي Boolean فهو عبارة Male و اما الحقل Number هو من النوع Age بينما الحقل 20 اي نص بطول Text نوعه Nameالحقل 

  .Falseاو 

توي بل يمكن لقاعدة البيانات تلك ان تح) ويمكن بالطبع ان يحتوي على اآثر ( Row يحتوي على حقلين Studentsو ان الجدول 
  .على اآثر من جدول

  . وانما هذا مثال للتوضيحXMLطبعا لا اريدك ان تفهم ان هذه بالضبط هي الطريقة التي يتم وصف قاعدة البيانات فيها بالـ 

 

----------------------------------------  

  من وجهة نظر المستخدم

  

 غير انها RSSهذه هي الـ ... فانني ببساطة اقول ! .. التي بدأت بها ؟اذا تسألت ما العلاقة بين عنوان الموضوع و المقدمة الطويلة 
   .2.0 تم الاتفاق عليها مسبقا ومنها اصدارات آثيرة لن اخوض فيها لانه ليس هدفي و سأآتفي بالاصدار XMLعبارة عن 

 انتباهك و يجذبك فمثلا يستهويني لم تعد الانترنت آما آانت مسبقا فملايين الصفحات منتشرة حول العالم و المئات منها يلفت
...  من موقع مايكروسوفت آما يروقني الاطلاع على حالة الطقس في اماآن تهمني و غيرها الكثير MSDNآخر مقالات مجلة الـ 

 آل يوم منا من يهتم بأخبار الرياضة او السياسة او الاخبار الرياضية و بالتالي اذا جمعت آل اهتماماتك ستجد انه ينبغي عليك في
  !عشرة او حتى عشرون موقع يوميا بل و ربما بشكل متكرر في اليوم, ان تزور خمسة 

 فما هي إلا عبارة عن مقتطفات سريعة عن المواضيع المختلفة مع روابط الى المواضيع آاملة على RSSهنا تكمن الحاجة الى الـ 
  .الانترنت

 فأنك ستلاحظ قالب صغير في الشريط الجانبي للفيستا يعرض لك آخر  آما في حالتيVistaفمثلا اذا آنت تستخدم الوينذوز 
  : انظر الى الصورة التالية RSSالاخبار من مواقع تقدم خدمة الـ 

 



 

 و بالتالي اذا لفت انتباهي خبر معين فيكفي ان انقر عليه MSNBCففي المربع الاسود في الشاشة تراني استعرض الاخبار من 
  : أشخاص في مجمع تجاري على يد شاب، و بالتالي عند النقر عليه يظهر8بر قتل  لفت انتباهي خمثلا

  

 

  

أليس من الأفضل ان يقدمها على الموقع و بالتالي يمكن للزوار ان ... ذه ؟ خدمة آهبتقديم سيقول قائل لماذا يقدم هذا الموقع 
 القائمون على الموقع يخسرون الكثير من زوارهم فليس في الحقيقة قولا آهذا سيجعل! ... يتصفحوه عوضا عن هذه المقتطفات 

تخيل معي ان معظم المواقع تقدم .. الجميع يمتلكون دائما الوقت الكافي ومن يمتلك الوقت الان لتصفح موقعك لن يمتلكه غدا 
يات التي تحبها وبالتالي هل تعرف ماذا سيحصل سيكون بإمكانك ان تعرف عناوين المشارآات الجديدة في المنتد! ... هذه الخدمة 

سترى نتائج المباريات ) مثلاآالجزيرة (لة ضفتح المواضيع التي تحب المشارآة فيها، ستعرف آخر الاخبار من مواقع الاخبار المف
 اليوم الرياضية ستعرف اسعار العملات ستتابع عدد الزائرون اليوم في موقعك الذي تديره و الاضافات التي اضافوها او ربما ماذا آتب

عن لغة البرمجة التي تفضلها، ستأخذ فكرة عن حالة  MSDNمن جديد في سجل الزوار بموقعك، ستعرف آخر المقالات التقنية للـ 
و سيسرك مثلا ان تجري اتصلا هاتفيا ) اذا آنت مثلي تعمل ببلد بعيد عنهم(اليوم الطقس في المنطقة التي تسكن فيها عائلتك 

!! ... لترى الدهشة عليهم وهم يسئلونك من اخبرك عن الامطار التي عندنا؟!! .. ي تهطل عليهم الان تسألهم فيه الامطار الت
 ).لايعرفون ان برنامجا بسيطا نقل الخبر الى سطح مكتبي(

  

مة  فما ان تزور موقع يروق لك ثم ترى علاRSSلخدمة الـ  من آون الدعم يزداد  آمستخدمحقيقة لايوجد حصر لما يمكن ان تجنيه

 فقل حينها الحمد الله) (او ) (الخدمة 

 

  

لكي تشترك في خدمة ماعليك سوى زيارة الرابط الذي سيقترحه عليك الموقع وسيقوم المتصفح لديك بتسجيلها لديك للإطلاع 
  Mozilla و في الـ Microsoft Internet Explorer IE7هذه الصورة توضح شكلها في الـ .. عليها لاحقا 



Firefox:

  

  :مثل ...  من الموقع التي تقدمها  RSS من البرامج الخاصة بقراءة الـ في الواقع توجد العديد

http://www.rssreader.com/ 

http://www.feedreader.com/ 

http://www.deskshare.com/awr.aspx 

  

  .او يمكنك الاعتماد على متصفح الانترنت الذي تستخدمه آما سبق اعلاه

  

-----------------------------------------------------------------------------------------------------------  



  من وجه نظر المطور

  

، فلربما آانت المقدمة قد اسالت لعابك عن الـ ا ينتهي حديثي معك آمستخدم ليبدأ حديثي معك آمبرمج و مطور مواقعالى هن
XML و اعقبنا ان خدمة الـ RSS هي عبارة عن وصف بإستخدام الـ XML و متفق عليها فيا ترى آيف يبدو ملف بسيط من ملفات الـ 
RSS 2.0 الاصدار:   

  

<?xml version="1.0" encoding="UTF-8"?> 

<rss version="2.0"> 

    <channel> 

        <title>عنوان القناة</title> 

        <description>وصفها ويمكن ان تكون قناة رغيد الاخبارية</description> 

        <link>http://msdn2.microsoft.com </link> 

        <lastBuildDate>Mon, 24 Sep 2007 07:33:37 -0700</lastBuildDate> 

        <pubDate>Wed, 21 Nov 2007 14:12:12 -0700 </pubDate> 

        <item> 

            <title>الخبر الاول</title> 

            <description>مجرد خبر ساذج لغرض التجربة</description> 

            <link>http://www.a.b.c.com</link> 

            <pubDate>Wed, 21 Nov 2007 14:12:12 -0700</pubDate> 

        </item> 

        <item> 

            <title>هذا موضوع ثاني</title> 

            <description>موضوع جميل ايضا لافائدة منه</description> 

            <link>http://www.anywhere.com</link> 

            <pubDate>Wed, 31 Oct 2007 17:34:15 -0700</pubDate> 

        </item> 

    </channel> 

</rss> 

اي جميع  (RSSفي الحقيقة قبل ان اختار وضع مثال فكرت مليا هل من المفيد ان اخوض في جميع قيم و احتملات ملف الـ 
  !!ام اآتفي بلمحة بسيطة صغيرة ؟) العناصر و خصائصها 



ت و يمكنك منها ان تنطلق الى ابعد ارتأيت اخيرا ان يكون المثال و ما يتبعه مقتصرا على القيم الشائعة و التي ستجدها اينما ذهب
  .من ذلك بكثير

  

ويجب ان يحتوي هذا العنصر على الخاصية  <rss> يجب ان يحتوي على او عنصر وهو RSSمن مثالنا يمكن القول بأن ملف الـ  
version التي تحدد الاصدار بالشكل <rss version=”2.0”> دمه في وهو ما سنستخ( هذا لإننا نستخدم الاصدار الثاني
  ).أمثلثنا

..  وهو العنصر الذي سيحتوي على جميع بيانات الخدمة <channel>على العنصر الاساسي  <rss>ويجب ان يحتوي عنصر الـ 
  :بالشكل التالي 

  

<channel> 

        <title> </title> 

        <description> </description> 

        <link> </link> 

        <lastBuildDate> </lastBuildDate> 

        <pubDate> </pubDate> 

        <item> 

  . . . . .   

        </item> 

</channel> 

 

 وهو Descriptionان القناه و  وفيه عنوTitle حتى نصفه وهي <channel> ومنها نرى ان عناصر فرعية يمكن اضافتها الى العنصر
 وهو تاريخ إنشاء القناة فإذا PubDateو وهو تاريخ التحديث  LastBuildDate وهو رابط الى موقع القناة الاخبارية و Linkوصفها و 

عنصر آنت تنشىء القناة بشكل يومي فينبغي عليك تغييره عند بداية آل يوم واذا حصل وقمت بتحديث شيء ما خلال اليوم غير ال
LastBuildDateوهو العنصر الخاص بالتحديث .  

  :وهو العنصر الذي يحتوي على الخبر بالشكل التالي  <item>واخيرا العنصر الاآثر اهمية 

  

<item> 

        <title>العنوان</title> 

        <description>الوصف</description> 

        <link>الرابط</link> 

        <pubDate>التاريخ</pubDate> 

 </item> 

 
 



على آثر من  <channel> و يمكن ان يحتوي العنصر ) العنوان و الوصف و الرابط و تاريخ الإنشاء (ستجد انه يحتوي على عناصر 
  .و ذلك بحسب الاخبار التي تريدها في القناة <item>عنصر من النوع 

اذا عدت الى الصورة الثانية في الموضوع ستجد انه ...  و هي امكانية احتوائها على روابط صور RSSاضافة جميلة الى الخدمة 
ت عليه فظهرت صورة ذلك الشاب وهذا نتيجة عندما لفت انباهي موضوع قتل شاب لثمانية اشخاص في مجمع تجاري نقر

  :ر الخاص بالصور وهو آالتاليصلإستخدام العن

 

  

<image> 

        <url>رابط الصورة</url> 

        <title>العنوان</title> 

        <link> ابط الموقعر </link> 

        <width>عرض الصور</width> 

        <height>ارتفاعها</height> 

        <description>وصفها</description> 

</image> 

 .لاحظ ان العناصر باللون الاحمر هي عناصر اختيارية يمكن الاستغناء عنها: ملاحظة

  .إليهما لكي تضاف صورة <item> أو <channel> من العنصرين أيا إلى <image> يضاف العنصر أنوبهذا يمكن 

  :للعلم فقط

 Item الخاص بكل Description و القيام بوضع الصور داخل الوصف imageهل العنصر  بتجاRSSيقوم الكثير من مزودي خدمة الـ 
  . فإذا قمت بعرضه يظهر على شكل نصوص و صور وهو أمر شائع نسبياHTMLبحيث يكون الوصف عبارة عن 

  

 بحيث تعرض لهم مثلا RSSة بخدم) سأسر آثيرا بكوني أول زبون( لنتخيل سويا أنك صاحب موقع إنترنت و تريد تزويد زبائنك ألانو 
 في موضوع معين بحيث يبقون مطلعين على آخر تحديثات الموضوع او ربما تريد عمل الإشراك إمكانية أو في موقعك الأخبارآخر 

آخر التعليقات الجديدة حتى تبقى على اطلاع ... إجمالي الزوار ! .. ألانخدمة لتزويدك بمعلومات عن موقع مثل عدد المتصفحين 
 التي تجري فيه بينما تقوم بتصفح الأمور فتحه في آل مرة فقد تصلك إلىئم بما يجري هناك على صفحات موقع دون الاضطرار دا

  !جريدتك الإخبارية اليومية

  

وبغض النظر عن نوايك فما تحتاج اليه هو إتصال بقاعدة بيانات تحتوي على معلومات ولنقل جدول فيه مجموعة من البيانات و 
 وبضعة سطور برمجية مثلا قاعدة بيانات Repeaterو بالتالي ستحتاج إلى ) ... العنوان و النص و التاريخ و الرابط ( مثل الحقول
 يقوم بإسترجاع البيانات إلى داخل SQL ثم امر OleDbConnectionإتصال  فنقوم بتكوين test و جدول بإسم data.mdbبإسم 

OleDbDataReaderلبيانات الـ  نجعله اخيرا آا مصدر Repeater.  

  

 و ما شابه و لكن لا بأس فيمكن بسهولة Repeater فلن تستطيع معرفة ما اتحدث عنه من ASP.NETحقيقة إذا لم تكن مبرمج 
 الكلاسيكية ASP أو الـ PHP أو لمبرمجي الـ #C بإي لغة تطوير و يمكن ان اضع امثلة لاحقاً بإذن االله تعالى بالـ RSSإنشاء خدمة 

   :ASP.NETو سأآتفي في هذا المقام بالـ ... ضا إي

 



    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs)  
 
   Dim CnnStr As String = "Provider=Microsoft.Jet.OleDB.4.0; Data Source=" 
        Dim Cnn As New OleDbConnection( CnnStr & Server.MapPath("data.mdb")) 
        Dim Cmd As New OleDbCommand("SELECT * FROM test", Cnn) 
 
        Try 
            Cnn.Open() 
            Dim rad As OleDbDataReader   
 

rad = Cmd.ExecuteReader(Data.CommandBehavior.CloseConnection) 
 
            Me.Repeater1.DataSource = rad 
            Me.Repeater1.DataBind() 
 
            rad.Close() 
        Catch ex As Exception 
        Finally 
            Cmd.Dispose() 
            If Cnn.State <> Data.ConnectionState.Closed Then Cnn.Close() 
        End Try 
 
    End Sub 

 

 

  !! ... في موقعك ، مبارك RSSونرى منه بوضوح ان الأمر بسيط جدا فللتو قمت بتكوين خدمة 

 : على الصفحة بشكل مشابه Repeaterصحيح لا تنسى ان تضع الـ 

  

<rss version="2.0"> 
    <channel> 
        <title>قناة رغيد الأخبارية</title> 
        <description>قناة لا تحتوي على شيء مفيد</description> 
        <link>http://msdn2.microsoft.com </link> 
        <lastBuildDate>Mon, 24 Sep 2007 07:33:37 -0700</lastBuildDate> 
        <pubDate>Wed, 21 Nov 2007 14:12:12 -0700 </pubDate> 
<asp:repeater id="Repeater1" runat="server"> 
    <ItemTemplate> 
        <item> 
            <title><%#DataBinder.Eval(Container.DataItem, "title")%></title> 
            <description><%#DataBinder.Eval(Container.DataItem, "desc")%></description> 
            <link><%#DataBinder.Eval(Container.DataItem, "desc")%></link> 
            <pubDate><%#DataBinder.Eval(Container.DataItem, "posttime")%></pubDate> 
        </item> 
    </ItemTemplate> 
</asp:repeater> 
    </channel> 
</rss> 
 
 

 



------------------------------------------------------------------------------  

  

  ! )مبرمجون مستهلكون (من وجهة نظر أمثالي 

 

و ) أمثالي( و سأتجه إلى المبرمجين المستهلكون RSSدون أن يقوموا بتزويد خدمة  الحديث مع المبرمجون الذين يريأطيللن 
   !!لإستهلاآيسأساعدهم على تنمية هذا التوجه ا

 و إن آانت قليلة على المستوى العربي للإسف الشديد غير انني اآيد بإذن االله RSSآثيرة هي المواقع التي تزود خدمات الـ 
  .تعالى من آون هذا التوجه يزداد عربيا بحمد االله تعالى و نسأل االله ان نكون مساهمون في آل خير بإذنه سبحانه

  في احد برامجك، فماذا يلزمك حينها ؟ RSS خدمة الـ غلالباست تريد ان تقوم كتخيل ان

  

يلزمك حينها ان تقوم بكتابة آود خاص بتحميل بيانات الخدمة من عنوانها فمثلا لنفرض انك تريد الإشتراك بخدمة آخر المقالات من 
  : ان تقوم بتحميل البيانات من عنوان الخدمة و الذي هو فما يلزمك حينهاcom.CodeProject.wwwموقع 

  

1=cat?aspx.articlerss/webservices/com.codeproject.www://http  

  

  وتجدها في المسارWebClient  لفئة اباستخدام مثلا Visual Basic.NETق الـ يويمكن ذلك بعدة طرق عن طر 
System.Net.WebClient  الدالة وباستخدامDownloadString() يمكنك ان تقوم بتحميل آافية بيانات صفحة الخدمة إلى متغير 

  :نصي بشكل مشابه

 Dim webSite As New System.Net.WebClient() 
 Dim rssString, URL As String  
 

“=  URL 1=cat?aspx.articlerss/webservices/com.codeproject.www://http” 

  
 rssString = webSite.DownloadString(URL) 
 

 

 

 و يمكنك الان التعامل معها آما rssStringي المتغير في السطرين اعلاه ستكون لديك البيانات آلها من الموقع مباشرة مخزنة ف
  .تشاء

  :ملاحظة

 بحيث تقةم بالقراءة مباشرة من الموقع بشكل متواصل عنصر بعد عنصر حتى النهاية، xmlTextReaderيمكن بالطبع استخدام 
تظار حتى تنتهي من عملها و بالتالي  تقوم بالتحميل مرة واحدة من البداية إلى النهاية و يعني ذلك الإنWebClientحيث أن الـ 

  . ستكون اسرع قليلاxmlTextReaderالقراءة المتواصلة عن طريق الـ 

  

 rssStringيعني و بعد ان يكون لدينا آل ما نريد في المتغير ... و سأستخدم انا في مثالي هنا الطريقة الاولى لسهولة الفهم 
 – و التي ذآرناها في بداية المقال RSSالبيانات اعني عناصر الـ (يانات منه نستطيع ان نقوم بتحليله على مهل لإستخراج الب

  ) .RSSلاتنسى ترآيب ملف الـ 

  



  :بالشكل التالي ...  بحيث يسهل التحرك خلاله XMLويمكن ايضا لتسهيل العمل تحميل النص آاملا إلى ملف 

 Dim webSite As New System.Net.WebClient() 
 Dim rssString, URL As String  
 

“=  URL 1=cat?aspx.articlerss/webservices/com.codeproject.www://http” 

  
 rssString = webSite.DownloadString(URL) 
 
 Dim doc As New XmlDocument() 
 doc.LoadXml(rssString) 
 
 

 

 

 فيه آل بياناتنا وما تبقى علينا هو دوران بسيط خلال هذا الملف xmlDocumentل البيانات في ملف فقد حملنا آ... والان فالتفرح 
 :بشكل مشابه للتالي ) rss,Channel,Item,Image( السابقة الذآر مثل RSS الموجودة فيه و إستخراج عناصر الـ Nodesو الـ 

  

 Dim webSite As New System.Net.WebClient() 
 Dim rssString, URL As String  
 

“=  URL 1=cat?aspx.articlerss/webservices/com.codeproject.www://http” 

  
 rssString = webSite.DownloadString(URL) 
 
 Dim doc As New XmlDocument() 
 doc.LoadXml(mXMLSource) 
 
 Dim Channel As XmlNode = Nothing 
 
 ' Look for the Channel Element. 
 For Each node As XmlNode In doc.ChildNodes 
     If node.Name.ToLower = "rss" Then 
  ... 
 ElseIF node.Name.ToLower = "channel" Then 
  ... 
 Next 
 .... 

 

 

يمكنك استغلال الفئات التي ... و لكي لا اجعل رأسك يصاب بالصداع .. ليك بإذن االله تعالى الامر سيكون بعدها سهل ع
  .سنناقشها سريعا في الفقرة التالية لتصميم برامجك و تطويرها آما تشاء



  

 rtaRSSنظرة سريعة على فئات الـ 

  

  

  !  جاءت ؟اما الاحرف الثلاة الاولى في إسم الفئات فلا تسألني عنها فلا اعرف من اي

 و اما الاحرف rtaRSS و اسم الفئات هو  Ragheed Al-Tayeb Ahmed ) رغيد الطيب أحمد أو (خاصة اذا عرفت أن اسمي هو  
  )؟ !سوف اصاب بإحباط شديد إذا بدأت بالتساؤل عن معناها  ( RSSالبقية فلا اظنها تخفى عليك 

  

 و يمكن استخدامهن لإعطائك rtaRSSImage و الثالثة rtaRSSItemانية  و الثrtaRSSChanelعلى العموم هن ثلاث فئات الاولى 
 دون عنا آتابة الكود الكثير فلكي تقوم بتحميل آل بيانات خدمة المقالات الاخيرة من موقع الـ RSSوصول سريع لخدمات الـ 

CodeProject فيمكنك ان تقوم بالتالي :  

  

 Dim rssChannel As New rtaRSSChannel() 
 Dim rssItem As New rtaRSSItem 
 
 rssChannel.URL = "http://www.codeproject.com/webservices/articlerss.aspx?cat=1" 
 rssChannel.UpdateNow() 
 
 For Each rssItem In rssChannel.Items 
     Dim Title As String = rssItem.Properties("title") 
     ListBox1.Items.Add(Title) 
 Next 
 

  

 التي ListBox1 في القائمة CodeProjectالمقالات الأخيرة في موقع كود السابق هو آل ما يلزمك لرؤية آل عناوين في الحقيقة ال
  !في برنامجك 

  

   rtaRSSChannelو لبيان الفئات دعني اذآر بسرعة اهم عناصر الفئة الرئيسية في مشروعي و هي 

  

  . التي تريد الارتباط بهاRSSأو بالاصح عنوان خدمة الـ  عنوان الموقع URLتحتوي هذه الفئة على خاصية مهمة هي 

 xmlSourceالخاصية 

  ).المصدر(و تحتوي على النص آاملا الذي تم تحميله من الموقع 

  Updatingالخاصية 

 .)Ready( تعني لاشيءFalse يعني انه جاري التحديث أو التحميل من الانترنت و Trueوهي عبارة عن خاصية لفحص حالة الفئة 

 Itemsالخاصية 

  . وسيتم وصفها لاحقاrtaRSSItemوهي الاهم حيث تعيد مصفوفة بكل العناصر التي تم تحميلها وهي من النوع 

  Imageالخاصية 

  .rtaRSSImageوهي من النوع ) في حال وجودها (channelتعيد الصورة المرتبطة بالقناة 



  LastErrorالخاصية 

  .لة اثنا التحميل أو خلافة ويمكنك استغلالها لتشخيص المشاآلوتعيد وصف آخر خطأ حصل داخل الدا

  Propertiesالخاصية 

  . وهكذا بالنسبة لبقية الخواصRSS آما جاء في وصف ملف الـ titleوتعيد آل خواص الفئة مثلا اذا اردت عنوان القناة فأستخدم الـ 

  

  :على اثنتين مهمتين هما  تحتوي Methodsوالان ننتقل إلى الحديث عن الدوال أو الاوامر 

 UpdateNowالأمر 

  .  في حالة نجاح العمليةTrueوهو المستخدم في المثال السابق و الذي يقوم بتحميل البيانات الخاصة بالخدمة من الانترنت ويعيد 

 Updateالأمر 

 مستقل Threadما يبدأها في يقوم بنفس عمل الامر السابق غير انه ليس متزامن اي انه لا ينتظر حتى تتنفذ العملية و إن
  في حال الفشل يفجر الحدث Updatedليسمح لك بإآمال شفرتك و في حال نجاع العملية وتحميل البيانات يفجر الحدث 

UpdateError.   

  : و بالتالي ففي حال انك اردت استخدامه سيتحول مثالك السابق إلى 

 Private WithEvents rssChannel As New rtaRSSChannel() 
 
 Private Sub Button1_Click(ByVal sender As System.Object, ByVl e As System.EventArgs) 
     rssChannel.URL = "http://www.codeproject.com/webservices/articlerss.aspx?cat=1" 
     rssChannel.Update() 
 End Sub 
 
 Private Sub rssChannel_Updated(ByVal Sender As Object) Handles rssChannel.Updated 
     For Each Item As rtaRSSItem In rssChannel.Items 
         Me.ListBox1.Items.Add(Item.Properties("title").ToString) 
     Next 
 End Sub 
 
 Private Sub rssChannel_UpdateError(ByVal Sender As rtaRSSChannel, ByVal ErrorString 
As String) Handles rssChannel.UpdateError 

MsgBox( "Error: " & ErrorString ) 
 End Sub 
 

 

 في حال نجاح Updated و الذي بدوره سيقوم بإستدعاء اما الحدث Update سيتم إستدعاء الأمر Button1فعند الضغط على الزر 
  . في حال العكسUpdateErrorالعملية أو 

  

 في حال آونك تريد استخدام احداث هذه الفئة آما جاء في WithEventsلا يبغي ان انبهك بضرورة الحرص على استخدام الـ 
  .كود اعلاهالسطر الاول من ال

Private WithEvents rssChannel As New rtaRSSChannel() 



 rtaRSSItemو اما الفئة 

 

 و الذي يحتوي على عنوان القناة الذي يتواجد هذا العنصر فيها آما ان ChannelTitleفلايوجد الكثير لأتحدث عنه فهناك الخاصية 
  . و خلافهtitle, description, pubDate, link جميع خصائص العنصر آالـ  و التي تحتوي علىPropertiesاهم خصائصه هي الخاصية 

  

و المهم ) يمكنك ان تختبرها و تفيدني بالنتائج( وحقيقة هي فئة لم اختبرها بعد rtaRSSImageو آذلك الحال بالنسبة للفئة 
 بسهولة أو يمكنك تطويرها RSSخدم خدمة الـ في الموضوع هو ان الفئة لازالت فئة ناضجة ولكن يمكنك استخدمها لبناء برامج تست

  .ايضا آما تشاء

  

  .ستجد مرفقا مع هذه المقالة مثال بسيط يستخدم هذه الفئات بإذن االله تعالى لعله يفيدك

  

 قبل أن اترآك اخبرك بانه سيكون لنا لقاء بإذن االله تعالى مع برنامج بسيط بنيته اعتمادا على هذه الفئة البسيطة يقوم بتصفح
  . التي قمت بالاشتراك بها واستعراضها بالشكل الذي يروقني، وفي الاسفل بعض اللقطات السريعة لهRSSخدمات الـ 

  

  



  

  

  

  

الحال اذا قمت بتفعيل الخدمة حقيقة لقد اطلت آثيرا في الموضوع ولكن سيكون من الجميل ان اضع آخر صورة آمثال عن 
 – الفريق العربي للبرمجة نقل آمثال اذا اردت ان اراقب المواضيع الجديدة في موقع منتدىبالنسبة للمنتديات التي تشارك فيها و ل

 Visual Basic .NET قسم الـ



  : لذاك القسم وهي RSSما احتاج إلى معرفته هو مسار خدمة الـ 

26=id&rssout=act?php.index/com.forum-2000rabteama.www://http  

  

  

  

  

  .مع تطبيقات لهذه الخدمة و إستغلالها في برامج متنوعة بإذن االله تعالى) ربما لقائات آخرى (لنا لقاء 

  

  

السلام عليكم و رحمة االله و برآاته



 rtaRSSوهي الـ  .. RSSالكود الكامل للفئات التواصل بخدمة الـ 

 

Imports System.Xml 
Imports System.Net 
 
Class rtaRSSChannel 
 
#Region "   Declaraction     " 
 
    Public Event Updated(ByVal Sender As rtaRSSChannel) 
    Public Event UpdateError(ByVal Sender As rtaRSSChannel, ByVal ErrorString As 
String) 
 
    Sub New() 
 
    End Sub 
 
    Sub New(ByVal URL As String) 
        mUrl = URL 
    End Sub 
 
#End Region 
 
#Region "   Public Methods   " 
 
    Function UpdateNow() As Boolean 
        mUpdating = True 
 
        Try 
            Dim web As New System.Net.WebClient() 
            Dim xmlString As String = web.DownloadString(mUrl) 
 
            mXMLSource = xmlString.Substring(xmlString.IndexOf("<rss")) 
 
            Return ReadXMLSource() 
        Catch ex As Exception 
            mUpdating = False 
            mLastError = ex.Message 
            RaiseEvent UpdateError(Me, ex.Message) 
            Return False 
        End Try 
    End Function 
 
    Sub Update() 
        Dim Trd As New Threading.Thread(AddressOf StartDownloading) 
        mUpdating = True 
        Trd.Start() 
    End Sub 
 
#End Region 
 
#Region "   Private Methods  " 
 
    Private Sub StartDownloading() 
        Try 
 
            Dim web As New System.Net.WebClient() 
            Dim xmlString As String = web.DownloadString(mUrl) 
            Dim r As New Xml.XmlTextReader("d") 
 
            mXMLSource = xmlString.Substring(xmlString.IndexOf("<rss")) 
 
            mUpdating = False 
 
            If ReadXMLSource() Then 



                RaiseEvent Updated(Me) 
            Else 
                RaiseEvent UpdateError(Me, mLastError) 
            End If 
 
        Catch ex As Exception 
            mUpdating = False 
            mLastError = ex.Message 
            RaiseEvent UpdateError(Me, ex.Message) 
        End Try 
    End Sub 
 
    Private Function ReadXMLSource() As Boolean 
        Try 
            mItems.Clear() 
            mProperties.Clear() 
            mImage = New rtaRSSImage 
 
            Dim doc As New XmlDocument() 
            doc.LoadXml(mXMLSource) 
 
            Dim Channel As XmlNode = Nothing 
 
            ' Look for the Channel Element. 
            For Each node As XmlNode In doc.ChildNodes 
                If node.Name.ToLower = "rss" Then 
                    For Each cnl As XmlNode In node.ChildNodes 
                        If cnl.Name.ToLower = "channel" Then 
                            Channel = cnl 
                            Exit For 
                        End If 
                    Next 
                    Exit For 
                End If 
            Next 
 
            ' Go into its items. 
            If Channel IsNot Nothing Then 
 
                Dim feed As rtaRSSItem 
                Dim title As String = "" 
 
                For Each element As XmlNode In Channel.ChildNodes 
                    Select Case element.Name.ToLower 
                        Case "item" 
                            feed = New rtaRSSItem(element) 
                            feed.ChannelTitle = title 
                            mItems.Add(feed) 
                        Case "image" 
                            mImage = New rtaRSSImage(element) 
                        Case Else 
                            If element.Name.ToLower = "title" Then 
                                title = element.InnerText 
                            End If 
 
                            If Not mProperties.Contains(element.Name) Then 
                                mProperties.Add(element.Name, element.InnerText) 
                            End If 
                    End Select 
                Next 
 
            End If 
 
            Return True 
        Catch ex As Exception 
            mLastError = ex.Message 
            Return False 
        End Try 



    End Function 
 
#End Region 
 
#Region "   Public Properties" 
 
    Private mUrl As String = "" 
    Property URL() As String 
        Get 
            Return mUrl 
        End Get 
        Set(ByVal value As String) 
            mUrl = value 
        End Set 
    End Property 
 
    Private mXMLSource As String 
    Property XMLSource() As String 
        Get 
            Return mXMLSource 
        End Get 
        Set(ByVal value As String) 
            mXMLSource = value 
            ReadXMLSource() 
        End Set 
    End Property 
 
    Private mTag As Object 
    Property Tag() As Object 
        Get 
            Return mTag 
        End Get 
        Set(ByVal value As Object) 
            mTag = value 
        End Set 
    End Property 
 
    Private mLastError As String = "" 
    ReadOnly Property LastError() As String 
        Get 
            Return mLastError 
        End Get 
    End Property 
 
    Private mProperties As New Hashtable 
    ReadOnly Property Properties() As Hashtable 
        Get 
            Return mProperties 
        End Get 
    End Property 
 
    Private mItems As New List(Of rtaRSSItem) 
    ReadOnly Property Items() As List(Of rtaRSSItem) 
        Get 
            Return mItems 
        End Get 
    End Property 
 
    Private mImage As New rtaRSSImage 
    ReadOnly Property Image() As rtaRSSImage 
        Get 
            Return (mImage) 
        End Get 
    End Property 
 
    Private mUpdating As Boolean = False 
    ReadOnly Property Updating() As Boolean 
        Get 



            Return mUpdating 
        End Get 
    End Property 
#End Region 
 
End Class 
 
 
Class rtaRSSItem 
    'Implements 
 
    Sub New() 
 
    End Sub 
 
    Private mChannelTitle As String = "" 
    Property ChannelTitle() As String 
        Get 
            Return (mChannelTitle) 
        End Get 
        Set(ByVal value As String) 
            mChannelTitle = value 
        End Set 
    End Property 
 
    Sub New(ByVal Node As XmlNode) 
        If Node.NodeType = XmlNodeType.Element AndAlso Node.Name.ToLower = "item" Then 
            For Each Item As XmlNode In Node.ChildNodes 
                If Item.Name.ToLower = "image" Then 
                    mImage = New rtaRSSImage(Item) 
                Else 
                    mProps.Add(Item.Name, Item.InnerText) 
                End If 
            Next 
        End If 
    End Sub 
 
    Private mProps As New Hashtable 
    ReadOnly Property Properties() As Hashtable 
        Get 
            Return mProps 
        End Get 
    End Property 
 
    Private mImage As New rtaRSSImage 
    ReadOnly Property Image() As rtaRSSImage 
        Get 
            Return mImage 
        End Get 
    End Property 
 
    Public Overrides Function ToString() As String 
        Return "rssItem: " & mProps("title") 
    End Function 
 
    Private mTag As Object 
    Property Tag() As Object 
        Get 
            Return mTag 
        End Get 
        Set(ByVal value As Object) 
            mTag = value 
        End Set 
    End Property 
End Class 
 
 
Class rtaRSSImage 



 
    Sub New() 
 
    End Sub 
 
    Sub New(ByVal Node As XmlNode) 
        If Node.NodeType = XmlNodeType.Element AndAlso Node.Name.ToLower = "image" 
Then 
            For Each Item As XmlNode In Node.ChildNodes 
                mProps.Add(Item.Name, Item.InnerText) 
            Next 
        End If 
    End Sub 
 
    Private mProps As New Hashtable 
    ReadOnly Property Properties() As Hashtable 
        Get 
            Return mProps 
        End Get 
    End Property 
 
    Public Overrides Function ToString() As String 
        Return "" 
    End Function 
 
End Class 

 

 

  ثم بحمد االله تعالى


